Example 8.7 Heat Conduction with an Insulator Boundary Condition
Example (8.1.17) is solved in Maple below:

```
> restart : with(inttrans) : with(plots) :
```

First, the governing equations and boundary conditions are converted to the Laplace domain and solved in the Laplace domain:

```
> eq:=diff(u(x,t),t)=diff(u(x,t),x$2);
```

$$eq := \frac{\partial}{\partial t} u(x, t) = \frac{\partial^2}{\partial x^2} u(x, t) \quad \textbf{(1)}$$

```
> u(x,0):=0;
```

$$u(x, 0) := 0 \quad \textbf{(2)}$$

```
> bc1:=diff(u(x,t),x)=0;
```

$$bc1 := \frac{\partial}{\partial x} u(x, t) = 0 \quad \textbf{(3)}$$

```
> bc2:=u(x,t)=1;
```

$$bc2 := u(x, t) = 1 \quad \textbf{(4)}$$

```
> eqs:=laplace(eq,t,s):
> eqs:=subs(laplace(u(x,t),t,s)=U(x),eqs);
```

$$eqs := s\, U(x) = \frac{d^2}{dx^2} U(x) \quad \textbf{(5)}$$

```
> bc1:=laplace(bc1,t,s):
> bc1:=subs(laplace(u(x,t),t,s)=U(x),bc1);
```

$$bc1 := \frac{d}{dx} U(x) = 0 \quad \textbf{(6)}$$

```
> bc2:=laplace(bc2,t,s):
> bc2:=subs(laplace(u(x,t),t,s)=U(x),bc2);
```

$$bc2 := U(x) = \frac{1}{s} \quad \textbf{(7)}$$

```
> dsolve(eqs,U(x));
```

$$U(x) = \_C1\, e^{\sqrt{s}\, x} + \_C2\, e^{-\sqrt{s}\, x} \quad \textbf{(8)}$$

```
> U(x):=c[1]*cosh(s^(1/2)*x)+c[2]*sinh(s^(1/2)*x);
```

$$U(x) := c_1 \cosh\left(\sqrt{s}\, x\right) + c_2 \sinh\left(\sqrt{s}\, x\right) \quad \textbf{(9)}$$

```
> eq0:=eval(subs(x=0,bc1));
```

$$eq0 := c_2 \sqrt{s} = 0 \quad \textbf{(10)}$$

```
> eq1:=eval(subs(x=1,bc2));
```

$$eq1 := c_1 \cosh\left(\sqrt{s}\right) + c_2 \sinh\left(\sqrt{s}\right) = \frac{1}{s} \quad \textbf{(11)}$$

```
> con:=solve({eq0,eq1},{c[1],c[2]});
```

$$con := \left\{ c_1 = \frac{1}{\cosh(\sqrt{s})\, s},\ c_2 = 0 \right\} \tag{12}$$

The solution obtained in the Laplace domain is:

```
> U(x):=subs(con,U(x));
```

$$U(x) := \frac{\cosh(\sqrt{s}\, x)}{\cosh(\sqrt{s})\, s} \tag{13}$$

The polynomials are:

```
> P(s):=numer(U(x));
```

$$P(s) := \cosh(\sqrt{s}\, x) \tag{14}$$

```
> Q(s):=denom(U(x));
```

$$Q(s) := \cosh(\sqrt{s})\, s \tag{15}$$

Note that the order of q(s) is greater than the order of p(s).

```
> A(s):=P(s)/diff(Q(s),s);
```

$$A(s) := \frac{\cosh(\sqrt{s}\, x)}{\frac{1}{2} \sinh(\sqrt{s})\, \sqrt{s} + \cosh(\sqrt{s})} \tag{16}$$

The roots of Q(s) are found as:

```
> solve(Q(s),s);
```

$$-\frac{1}{4}\pi^2,\ 0 \tag{17}$$

```
> _EnvAllSolutions := true;
```

$$\_EnvAllSolutions := true \tag{18}$$

```
> solve(Q(s),s);
```

$$-\frac{1}{4}\pi^2 (1 + 2\,\_Z1\sim)^2,\ 0 \tag{19}$$

The roots can be taken as:

```
> 0,-((2*n-1)*Pi/2)^2;
```

$$0,\ -\frac{1}{4}(2n-1)^2\pi^2 \tag{20}$$

Next, the coefficients are found:

```
> A[n]:=simplify(subs(s=mu,A(s)));
```

$$A_n := \frac{2\cosh(\sqrt{\mu}\, x)}{\sinh(\sqrt{\mu})\, \sqrt{\mu} + 2\cosh(\sqrt{\mu})} \tag{21}$$

First A0 is found:

```
> A[0]:=subs(mu=0,A[n]);
```

$$A_0 := 1 \tag{22}$$

The coefficient An for values n = 1..∞ can be found as:

```
> A[n]:=simplify(subs(mu^(1/2)=I*(2*n-1)/2*Pi,A[n]));
```

$$A_n := -\frac{4\cos\left(\frac{1}{2}\,(2\,n-1)\,\pi\,x\right)}{2\sin\left(\frac{1}{2}\,(2\,n-1)\,\pi\right)\pi\,n-\sin\left(\frac{1}{2}\,(2\,n-1)\,\pi\right)\pi-4\cos\left(\frac{1}{2}\,(2\,n-1)\,\pi\right)} \qquad \textbf{(23)}$$

A_n is simplified as:

```
> vars:={cos(1/2*(2*n-1)*Pi)=0,sin(1/2*(2*n-1)*Pi)=(-1)^(n-1)};
```

$$vars := \left\{\cos\left(\frac{1}{2}\,(2\,n-1)\,\pi\right)=0,\ \sin\left(\frac{1}{2}\,(2\,n-1)\,\pi\right)=(-1)^{n-1}\right\} \qquad \textbf{(24)}$$

```
> A[n]:=simplify(subs(vars,A[n]));
```

$$A_n := \frac{4\,(-1)^{-n}\cos\left(\frac{1}{2}\,(2\,n-1)\,\pi\,x\right)}{\pi\,(2\,n-1)} \qquad \textbf{(25)}$$

The general terms in the Laplace domain solution are (see equation (8.1.16)):

```
> u0s:=A[0]*1/s;
```

$$u0s := \frac{1}{s} \qquad \textbf{(26)}$$

The inverse Laplace transform is:

```
> u0t:=invlaplace(u0s,s,t);
```

$$u0t := 1 \qquad \textbf{(27)}$$

The term in the infinite series is

```
> uns:=A[n]/(s-mu);
```

$$uns := \frac{4\,(-1)^{-n}\cos\left(\frac{1}{2}\,(2\,n-1)\,\pi\,x\right)}{\pi\,(2\,n-1)\,(s-\mu)} \qquad \textbf{(28)}$$

```
> unt:=invlaplace(uns,s,t);
```

$$unt := \frac{4\,(-1)^{-n}\cos\left(\frac{1}{2}\,(2\,n-1)\,\pi\,x\right)e^{\mu t}}{\pi\,(2\,n-1)} \qquad \textbf{(29)}$$

```
> unt:=subs(mu=-((2*n-1)/2*Pi)^2,unt);
```

$$unt := \frac{4\,(-1)^{-n}\cos\left(\frac{1}{2}\,(2\,n-1)\,\pi\,x\right)e^{-\frac{1}{4}(2\,n-1)^2\pi^2 t}}{\pi\,(2\,n-1)} \qquad \textbf{(30)}$$

The final solution is obtained as:

```
> U:=u0t+Sum(unt,n=1..infinity);
```

$$U := 1+\sum_{n=1}^{\infty}\frac{4\,(-1)^{-n}\cos\left(\frac{1}{2}\,(2\,n-1)\,\pi\,x\right)e^{-\frac{1}{4}(2\,n-1)^2\pi^2 t}}{\pi\,(2\,n-1)} \qquad \textbf{(31)}$$

As in chapter 7, the initial condition is used at time, t = 0, to avoid Gibb's phenomenon:

```
> u:=piecewise(t=0,0,t>0,subs(infinity=20,U));
```

$$u := \begin{cases} 0 & t=0 \\ 1+\displaystyle\sum_{n=1}^{20} \frac{4\,(-1)^{-n}\cos\left(\frac{1}{2}\,(2\,n-1)\,\pi\,x\right)\mathrm{e}^{-\frac{1}{4}\,(2\,n-1)^2\,\pi^2\,t}}{\pi\,(2\,n-1)} & 0<t \end{cases} \tag{32}$$

The following plots are obtained:

```
> plot3d(u,x=0..1,t=0..1,axes=boxed,title="Figure Exp. 8.13.",
  labels=[x,t,"u"],orientation=[-150,60]);
```

Figure Exp. 8.13.




```
> plot([subs(t=0,u),subs(t=0.1,u),subs(t=0.2,u),subs(t=0.3,u)],x=
  0..1,axes=boxed,title="Figure Exp. 8.14.",thickness=5,labels=[x,
  "u"]);
```

Figure Exp. 8.14.



>